2000 г. Труды ФОРА №5

ПРИМЕНЕНИЕ МАТЕМАТИЧЕСКОЙ СРЕДЫ MAPLE В ТЕОРЕТИЧЕСКОЙ ФИЗИКЕ

П. Г. Коваль

Адыгейский государственный университет, Майкоп

В статье рассмотрено несколько примеров использования математической среды Maple при рассчётах по теоретической физике. Приведены фрагменты конкретных программ. Проведено сравнение среды с другими аналогичными продуктами.

Введение

Современная физика развивается в двух направлениях. Первое направление это усложнение математического аппарата, развитие передовых областей математики и их применение к исследованию физических проблем. Именно этим путём в физику вошли дифференциальные уравнения и интеграл в 19 веке, тензорное исчисление, матрицы и теория групп в 20 веке. Некоторые законы современной физики могут быть описаны с помощью аппарата дифференциальных уравнений. Сложность решения таких уравнений, сложность получаемых ответов дала толчок к развитию численных методов, которые являются, по сути, моделями дифференциальных уравнений.

Второй путь развития физики состоит в усложнении численных методов. При этом принципы, сформулированные через исчисление бесконечно малых, сомнению не подвергаются. Численные методы используются для решения сложных, но конкретных задач, поэтому результаты, получаемые в каждом вычислении носят частный характер. Развитие вычислительной техники сделало возможным проведение численных экспериментов при широком изменении параметров задачи на основе общих дифференциальных принципов. Результаты, получаемые при такого рода экпериментировании являются, по сути, результатами "из первых рук", и в некоторых случаях это единственные результаты, которые сравниваются с экспериментом.

Очень недолго компьютеры помогали только в численных рассчётах. По мере увеличения производительности и объёмов памяти появилась возможность использовать компьютеры для аналитических выкладок. Для упрощения символьных выражений, нахождения аналитических решений уравнений. Поэтому компьютер – современный мощный инструмент – может быть использован как в численных рассчётах, так и в аналитических (символьных преобразованиях) [1], [2], [3].

Опыт использования математического программного обеспечения показывает, что для проведения численных рассчётов часто пишутся оригинальные программы, в то время как для аналитических рассчётов это не эффективно, потому что уже существуют программы, которые могут делать практически всё в математике. Выдумывать что-то своё в этой области означает изобретение плохого велосипеда.

Из всего разнообразия символьных процессоров выделяются несколько программ: Mathematica, Maple, MathCAD. Современные версии этих программ обладают сходными базовыми возможностями, но имеют различный интерфейс, различный синтаксис команд и набор дополнительных пакетов.

С нашей точки зрения математическая среда Maple (Мэйпл – Клён) является наиболее удобной в работе, неприхотливой в настройке и в то же время мощной математической средой. Были проведены и проводятся рассчёты, которые без использования Maple были бы признаны слишком сложными и могли остаться незавершёнными.

Простота освоения математической среды даёт возможность её использования в учебных целях. В Инттернет на сайте Санкт-Петербургского Государственного Университета можно получить добротный курс лекций по квантовой механике, полностью написанный в Maple.

Maple и пользователь

Maple это среда, которая содержит редактор и интерпретатор команд пользователя. Она идейно близка к известной системе REDUCE [4], [5]. Обладая большим потенциалом в математике, Maple имеет богатые графические возможности, которых не было в REDUCE. Для знакомых с REDUCE требуется минимальное переучивание, чтобы начать работу. Подавляющее большинство синтаксических конструкций одинаковы в обоих системах.

Команды имеют простой синтаксис и сходны с такими языками как C, Fortran, Pascal. Например, чтобы извлечь квадратный корень, можно набрать команду

```
> sqrt(10);
```

$$\sqrt{10}$$

```
>
```

после нажатия клавиши "ввод" мы видим, что Maple просто отобразил результат вычисления. Видно, что Maple не бросается вычислять корень, он различает $\sqrt{10}$ и 3.162277660. Если же нам нужен ответ в виде числа, то следует воспользоваться функцией

```
> evalf(sqrt(10));
```

$$3.162277660$$

evalf – это функция, которая вычисляет всё что можно вычислить с использованием арифметики с плавающей точкой (floating point arithmetic). Она является частью большого класса функций eval, которые вычисляют выражение с использованием дополнительных знаний. Часто используемыми оказались функции evalc и evalm: первая вычисляет всю комплексную часть выражения, а вторая вычисляет матричные выражения. Как это выглядит на практике видно из примера

```
> restart; A:=a+I*b; B:=a+I*b; evalc(A*B);
```

$$A := a + I\,b$$

$$B := a + I\,b$$

$$a^2 + 2\,I\,a\,b - b^2$$

```
> A:=matrix(2, 2, [[alpha, beta], [gamma, delta]]);
```

$$A := \begin{bmatrix} \alpha & \beta \\ \gamma & \delta \end{bmatrix}$$

```
> B:=matrix(2, 2, [[a, b], [c, d]]):  evalm(A&*B);
```

$$\begin{bmatrix} \alpha\,a + \beta\,c & \alpha\,b + \beta\,d \\ \gamma\,a + \delta\,c & \gamma\,b + \delta\,d \end{bmatrix}$$

из примера так же ясно, что Maple различает большие и малые буквы как язык C и может отображать греческие буквы в привычном виде.

После команды restart Maple начинает работать как только что запущенная программа. Это не лишняя предосторожность, так как все окна среды имеют единое вычислительное ядро, поэтому может случится так, что какой-либо символ обозначает совсем не то, что о нём можно подумать. Каждый символ в Maple может обозначать число, выражение, матрицу, вектор, равенство, неравенство, оператор, процедуру, список и ещё многое другое. Некоторые символы, например I, gamma, infinity, являются константами и обозначают вполне определённые математические понятия. Их лучше всего использовать по предопределённому назначению. В случае символа gamma, который изначально содержит константу Эйлера, это не всегда удобно, поэтому можно ему присвоить нужное выражение как это показано ниже

```
> restart; evalf(gamma); assume(gamma, real);
```

$$.5772156649$$


```
Error, (in assume) cannot assume on a constant object
```

```
> unprotect(gamma); type(gamma, constant); gamma:=beta1^2;
```

$$true$$

$$\gamma := \beta 1^2$$

```
> gamma; evalf(gamma); about(gamma);
```

$$\beta 1^2$$

$$\beta 1^2$$

```
beta1^2:
  nothing known about this object
```

Чтобы получить информацию о свойствах конкретного символа служит команда (функция) about. Чтобы определить свойства некоторого символа служит функция assume

```
> restart;
> assume(a>0); sqrt(a^2);
```

$$a$$

```
> sqrt(b^2);
```

$$\sqrt{b^2}$$

Видно, что Maple в случае переменной a упростил квадратный корень из квадрата, но не сделал этого для переменной b.

Maple содержит целый набор функций для упрощения и преобразования выражений. Это функции simplify, expand, combine, factor

```
> i1:=(n*GAMMA(n+alpha+1)^2/(n)!/GAMMA(n+alpha));
```

$$i1 := \frac{n\,\Gamma(n+\alpha+1)^2}{n!\,\Gamma(n+\alpha)}$$

```
> simplify(i1);
```

$$\frac{(n+\alpha)\,\Gamma(n+\alpha+1)}{\Gamma(n)}$$

```
> expand(i1);
```

$$\frac{n\,\Gamma(n+\alpha)\,(n+\alpha)^2}{n!}$$

из их названия можно понять что они делают. То выражение, которое эти функции выдают на выходе не всегда приемлемо, но всегда эквивалентно тому выражению, которое они преобразуют. Эти функции целесообразно использовать вместе с assume. Если выражение заканчивается двоеточием, то Maple проделывает все вычисления, но не пишет результат.

Подводя итоги, можно сказать, что Maple это программируемая высокоуровневая среда. Она содержит мощный символьный процессор, который работает в режиме интерпретатора команд и мощную систему визуализации результатов вычислений. Maple содержит так же конверторы своих рабочих листов (worksheet) в формат LaTeX и HTML (в версии Maple V Release 5.1 и старше), что делает её исключительно удобной системой.

Задачи, к которым была применена среда Maple

Коэффициенты Ньютона-Котеса

Одна из первых задач, к которым была применена среда, заключалась в вычислении коэффициентов Ньютона-Котеса. При численном интегрировании по методу Ньютона-Котеса заменяют подынтегральную функцию на многочлены, узлы которых равноотстоят друг от друга и значения в узлах совпадают со значениями подынтегральной функции. Ожидается, что при использовании многочленов всё больших степеней потребуется меньше арифметических операций для достижения заданной точности взятия интеграла.

Maple содержит стандартные средства для работы с конечными и бесконечными суммами и произведениями, она может так же брать интегралы и упрощать результат. Этого оказалось вполне достаточно, чтобы сделать программу, которая вычисляет коэффициенты Ньютона-Котеса N-го порядка, причём N – число, которое задаётся в начале вычислений. Ниже приведена эта программа.

Это формула второго порядка (формула Симпсона)

```
x1:=0: x2:=h: x3:=2*h:
f(x):=(x-x2)*(x-x3)*y1/(x1-x2)/(x1-x3)+(x-x1)*(x-x3)*y2/(x2-x3)/(x2-x1)
+(x-x1)*(x-x2)*y3/(x3-x1)/(x3-x2):
Int(f1(x), x=0..2*h)=factor(int(f(x), x=0..2*h));
```

$$\int_0^{2\,h} \mathrm{f1}(x)\,dx = \frac{1}{3}\,h\,(y1 + 4\,y2 + y3)$$

Ниже вычисляется формула заранее заданного порядка N

```
restart;
N:=9:
for k from 0 by 1 to N-1 do x[k+1]:=k*h od:
g1(x):=product(x-x[j], j=1..i-1)*product(x-x[j], j=i+1..N):
g2(x):=product(x[i]-x[j], j=1..i-1)*product(x[i]-x[j], j=i+1..N):
f(x):=sum(y[i]*g1(x)/g2(x), i=1..N):
collect(expand(int(f(x), x=0..(N-1)*h)), h);
```

$$\Big(\frac{41984}{14175}\,y_4 - \frac{3632}{2835}\,y_5 - \frac{3712}{14175}\,y_3 + \frac{3956}{14175}\,y_1 + \frac{23552}{14175}\,y_2 + \frac{3956}{14175}\,y_9 + \frac{23552}{14175}\,y_8 + \frac{41984}{14175}\,y_6 - \frac{3712}{14175}\,y_7\Big)h$$

Результаты можно сравнить с известными формулами [6], ([7], С. 683.) В программе использованы функции Int и int. int пытается взять интеграл, а Int обозначает тот же интеграл, но невычисленный.

Корни многочленов Эрмита

Maple может не только упрощать выражения, которые ему пишут, но и сам получать выражения из уравнений. Для этой цели служит функция solve. При решении уравнения Кляйна-Гордона для частицы в ступенчатом магнитном поле

$$\mathbf{H} = \{0,\ 0,\ H_z\};\ H_z = 2H_0(\mathbf{1}(x) - 1/2).$$

$$\mathbf{A} = \{0,\ A_y,\ 0\};\ A_y = 2H_0x(\mathbf{1}(x) - 1/2).$$

было необходимо "сшить" найденные волновые функции и их производные на границе $x = 0$ (здесь $\mathbf{1}(x)$ – функция Хевисайда). Сами решения, которые верны в каждой из областей $x < 0$ и $x > 0$ известны и представляют из себя полиномы Эрмита

$$\Psi_1(\xi) = A\mathrm{e}^{-\frac{\xi^2}{2}} H_n(\xi) \quad x > 0;$$
$$\Psi_2(\zeta) = B\mathrm{e}^{-\frac{\zeta^2}{2}} H_n(\zeta) \quad x < 0,$$

где $\xi = \sqrt{2\gamma}x + \frac{k_2}{\sqrt{2\gamma}}$, $\zeta = \sqrt{2\gamma}x - \frac{k_2}{\sqrt{2\gamma}}$, а $H_n(x)$ – полиномы Эрмита

$$H_n(x) = (-1)^n \mathrm{e}^{x^2} \frac{d^n \mathrm{e}^{-x^2}}{dx^n}.$$

Условие непрерывности волновой функции и её производной по координате приводит к ограничению на значения волнового числа k_2.

Приравнивая значения волновых функций $\Psi_1(\xi)$ и $\Psi_2(\zeta)$ при $x = 0$, получим уравнение:

$$H_n(k_2/\sqrt{2\gamma}) = H_n(-k_2/\sqrt{2\gamma}). \tag{1}$$

При чётных значениях индекса n полиномы Эрмита содержат только чётные степени аргумента, поэтому уравнение (1) удовлетворяется при любых значениях $\kappa_2 = k_2/\sqrt{2\gamma}$. При нечётных значениях n уравнение (1) приводит к нулям полиномов Эрмита

$$H_n(\kappa_2) = 0.$$

Приравнивая значения производных от волновых функций по координате x при $x = 0$, получаем уравнение

$$-\kappa_2 H_n(\kappa_2) + 2nH_{n-1}(\kappa_2) = \kappa_2 H_n(-\kappa_2) + 2nH_{n-1}(-\kappa_2), \tag{2}$$

которое автоматически удовлетворяется при нечётных значениях индекса n. Если n чётное, то уравнение (2) приводится к условию

$$\kappa_2 H_n(\kappa_2) = 2nH_{n-1}(\kappa_2).$$

Документ (worksheet) Maple, который использовался для решения, полученных уравнений приведён ниже.

```
> restart;

  H:=(n, xi)->simplify((-1)^n*exp(xi^2)*subs(%x=xi,
  diff(exp(-%x^2),}{%x$n)));
```

$$H := (n, \xi) \to \mathrm{simplify}((-1)^n\, e^{(\xi^2)}\, \mathrm{subs}(\%x = \xi,\, \mathrm{diff}(e^{(-\%x^2)},\, \%x\,\$\, n)))$$

Условие равенства производных.

```
> n:=4;
```

$$n := 4$$

```
> g1:=simplify(kappa2*H(n, kappa2)=2*n*H(n-1, kappa2));
```

$$g1 := 4\,\kappa 2\,(3 - 12\,\kappa 2^2 + 4\,\kappa 2^4) = 32\,\kappa 2\,(-3 + 2\,\kappa 2^2)$$

```
> evalf(solve(g1, kappa2));
```

$$0,\ 2.417686473,\ -2.417686473,\ 1.074612544,\ -1.074612544$$

Условие равенства функций.

```
> n:=3; g2:=H(n, kappa2); solve(g2, kappa2);
```

$$n := 3$$

$$g2 := 4\,\kappa 2\,(-3 + 2\,\kappa 2^2)$$

$$0,\ \frac{1}{2}\sqrt{6},\ -\frac{1}{2}\sqrt{6}$$

Здесь в первой выполняемой группе (execution group) определяется оператор H двух переменных. Это полином Эрмита n-го порядка. Чтобы иметь возможность использовать оператор не только для символов, но и для чисел была введена "внутренняя" переменная $\%x$, по которой и происходит дифференцирование. После дифференцирования все переменные $\%x$ заменяются на второй аргумент оператора H. Аргументом дифференцирования не может быть константа, в то время как операция замены (substitution) может заменить $\%x$ и константой. Недостатком оператора является то, что можно использовать только положительные n в первом аргументе. Далее переменной g1 присваивается равенство, которое решается командой solve.

Решение уравнения может быть не одно, поэтому все найденные решения записываются функцией solve в список. На выходном дисплее видно, что элементы списка разделены запятыми. Чтобы получить доступ к отдельному элементу списка можно использовать индекс в квадратных скобках.

```
> s1:=evalf(solve(g1, kappa2)); s1[1]; s1[2];
```

$$s1 := 0,\ 2.417686473,\ -2.417686473,\ 1.074612544,\ -1.074612544$$

$$0$$

$$2.417686473$$

Функция solve может быть использована для решения систем уравнений как это показано ниже

```
> restart; g1:=a1*x+b1*y+c1=0; g2:=a2*x+b2*y+c2=0;
```

$$g1 := a1\,x + b1\,y + c1 = 0$$

$$g2 := a2\,x + b2\,y + c2 = 0$$

```
> s1:=solve({g1, g2}, {x, y});
```

$$s1 := \{x = \frac{-b2\,c1 + c2\,b1}{a1\,b2 - a2\,b1},\ y = -\frac{a1\,c2 - c1\,a2}{a1\,b2 - a2\,b1}\}$$

```
> assign(s1); y; x;
```

$$-\frac{a1\,c2 - c1\,a2}{a1\,b2 - a2\,b1}$$

$$\frac{-b2\,c1 + c2\,b1}{a1\,b2 - a2\,b1}$$

В последних группах демонстрируется функция assign, которая символу в левой части равенства присваивает правую часть. Функция assign действует так же для списков равенств.

Дифференциальные уравнения

Наиболее удивительной является способность Maple решать дифференциальные уравнения, системы дифференциальных уравнений и уравнения в частных производных.

В процессе поиска решения уравнения Кляйна-Гордона для поля двух бесконечно тонких соленоидов было решено испытать систему координат, в которой касательные к координатным линиям везде параллельны вектор-потенциалу поля двух соленоидов. Иными словами надо было решить уравнение

$$\frac{dy}{dx} = \frac{A_y}{A_x},$$

которое даёт искомые координатные линии, и за новую переменную взять произвольную константу, которая определяет конкретную линию. Координатная система должна быть ортогональной. Из этого условия была найдена вторая искомая координата, Для чего нужно решить уравнение

$$\frac{dy}{dx} = -\frac{A_x}{A_y},$$

Описанный способ был реализован в программе Maple

```
> restart; x0:=d:  x1:=-x0:  x2:=+x0:  y1:=-x0:  y2:=+x0:
> A1:=[-A0*y/((x-x1)^2+y^2), A0*(x-x1)/((x-x1)^2+y^2)]:
> A2:=[-A0*y/((x-x2)^2+y^2), A0*(x-x2)/((x-x2)^2+y^2)]:
> A:=(A1+A2);
```

$$A := [-\frac{A0\, y}{(x-d)^2+y^2} - \frac{A0\, y}{(x+d)^2+y^2}, \frac{A0\,(x-d)}{(x-d)^2+y^2} + \frac{A0\,(x+d)}{(x+d)^2+y^2}]$$

```
> ds1:=diff(y(x), x)=subs(y=y(x), simplify(A[2]/A[1]));
```

$$ds1 := \frac{\partial}{\partial x}\,\mathrm{y}(x) = \frac{x\,(-x^2+d^2-\mathrm{y}(x)^2)}{\mathrm{y}(x)\,(x^2+d^2+\mathrm{y}(x)^2)}$$

```
> ds2:=diff(y(x), x)=subs(y=y(x), -simplify(A[1]/A[2]));
```

$$ds2 := \frac{\partial}{\partial x}\,\mathrm{y}(x) = -\frac{\mathrm{y}(x)\,(x^2+d^2+\mathrm{y}(x)^2)}{x\,(-x^2+d^2-\mathrm{y}(x)^2)}$$

```
> s1:=dsolve(ds1, y(x))[1];
```

$$s1 := \mathrm{y}(x) = \sqrt{-d^2-x^2+\sqrt{d^4+4\,x^2\,d^2-4\,\_C1}}$$

```
> s2:=dsolve(ds2, y(x))[1];
```

$$s2 := \mathrm{y}(x) = -\frac{1}{2}\,\_C1\,x + \frac{1}{2}\sqrt{\_C1^2\,x^2-4\,d^2+4\,x^2}$$

```
> u=subs(y(x)=y, solve(s1, _C1));
```

$$u = \frac{1}{2}\,x^2\,d^2 - \frac{1}{4}\,x^4 - \frac{1}{2}\,x^2\,y^2 - \frac{1}{2}\,y^2\,d^2 - \frac{1}{4}\,y^4$$

```
> v=subs(y(x)=y, solve(s2, _C1));
```

$$v = -\frac{d^2-x^2+y^2}{y\,x}$$

Уравнение $u = \mathrm{const}$, как оказалось, определяет обобщенную лемнискату. Maple справился с заданием, хотя это и не привело к решению задачи о двух соленоидах, потому что оставшаяся компонента вектор-потенциала A_u зависит от двух координат $A_u = A_u(u, v)$.

Maple содержит описания 30 трёхмерных систем координат и 16 двумерных. В дополнение к этому существует возможность вводить пользовательские системы координат, для которых Maple самостоятельно рисует координатные линии.

Графические возможности Maple

Maple содержит набор специализированных пакетов (package), которые можно с успехом использовать, подключив соответствующую библиотеку командой with. Наиболее часто используемыми являются библиотеки plots и linalg. Первая содержит в себе функции для представления данных в виде графиков и диаграмм. Вторая содержит набор функций для работы с матрицами и векторами.

Чтобы построить координатные линии $u = u_1$ и $v = v_1$ была использована функция implicitplot, позволяющая строить линии, заданные неявно. Продолжение предыдущего примера демонстрирует применение функций implicitplot и display (display для изображения нескольких линий на одном чертеже).


```
> u:=5; v:=2; with(plots):
```

$$u := 5$$

$$v := 2$$

```
> im2:=subs(d=2, -solve(subs(y(x)=y, s1), _C1))=u;
```

$$im2 := -2\,x^2 + \frac{1}{4}\,x^4 + \frac{1}{2}\,x^2\,y^2 + 2\,y^2 + \frac{1}{4}\,y^4 = 5$$

```
> im3:=subs(d=2, solve(subs(y(x)=y, s2), _C1))=v;
```

$$im3 := -\frac{4 - x^2 + y^2}{y\,x} = 2$$

```
> A1:=implicitplot(im2, x=-5..5, y=-5..5, view=[-5..5, -5..5]):
> A2:=implicitplot(im3, x=-5..5, y=-5..5, view=[-5..5, -5..5]):
> display(A1, A2);
```



Библиотека plots содержит функции для построения трёхмерных графиков и геометрических фигур, рисования векторных полей, построения диаграмм плотности (density plotting) и анимации. Ниже приведена программа, которая постоила векторное поле задачи о двух соленоидах.

```
> with(plots):  x1:=1:  x2:=-1:
> A1:=[-y/((x-x1)^2+y^2), (x-x1)/((x-x1)^2+y^2)]:
> A2:=[-y/((x-x2)^2+y^2), (x-x2)/((x-x2)^2+y^2)]:
> fieldplot(A1+A2, x=-2..2, y=-2..2, arrows=THICK);
```



Построение трёхмерных индикатрис интенсивности излучения

Движение было задано в параметрическом виде [8]

$$\mathbf{r} = \left\{ c\beta\gamma(\beta\xi + \cos\xi), c\beta\gamma\sqrt{1-\beta^2}\sin\xi, 0 \right\}; \quad t' = \gamma\xi + \gamma\beta\cos\xi, \tag{3}$$

где $\gamma = 1/\omega(1-\beta^2)$, ξ – параметр, β – модуль скорости частицы, делённый на скорость света.

Спектрально-угловое распределение интенсивности излучения рассчитывалось по классической формуле ([9], С. 230)

$$\overline{\frac{dI_n}{d\Omega}} = \frac{2\pi n^2 e^2}{c^3 T^4} \left| \int\limits_0^{2\pi} \left[\frac{d\mathbf{r}}{d\xi}\mathbf{n} \right] e^{-i\omega_0 n(t' - \mathbf{nr}/c)}\, d\xi \right|^2,$$

где аргумент экспоненты и векторное произведение будут

$$n\omega_0(t' - \mathbf{nr}/c) = n\left(\xi + \frac{\beta}{1 - n_x\beta^2} \left(\cos\xi(1-n_x) - \sin\xi\sqrt{1-\beta^2}\, n_y \right) \right).$$

$$\left[\frac{d\mathbf{r}}{d\xi}\mathbf{n} \right] = c\beta\gamma \left(n_z\sqrt{1-\beta^2}\cos\xi,\; n_z(\sin\xi - \beta), n_y\beta - n_y\sin\xi - n_x\sqrt{1-\beta^2}\cos\xi \right).$$

Интегралы брались численно методом Гаусса программой на Паскале. Она формировала текстовый файл с координатами трёхмерных треугольников.

```
d:=[[[x11, y11, z11], [x12, y12, z12], [x13, y13, z13]],
    [[x21, y21, z21], [x22, y22, z22], [x33, y33, z33]],
     ...
    [[xn1, yn1, zn1], [xn2, yn2, zn2], [xn3, yn3, zn3]]]:
```

Файл обрабатывался программой Maple.

```
with(plots):
read `c:/maple/3D71.txt`:
polygonplot3d(d, axes=framed, style=patch, shading=none,
labels=['x','y','z'], tickmarks=[5,3,0], orientation=[90,0],
lightmodel=none, scaling=constrained, projection=0.5);
```

В результате получались рисунки, показанные ниже



а) $n = 1$



б) $n = 2$

Из примера видно как записывать путь к файлу в Maple.

Движение частицы в поле плоских электромагнитных ударных волн

В плоских ударных электромагнитных волнах частицы ускоряются, т. е. после взаимодействия с ударной волной частица получает дополнительную энергию.

Уравнение движения частицы в классическом случае для плоской волны, которая распространяется вдоль оси OZ, даёт решение

$$p_z = \frac{e^2}{mc\omega^2\alpha}\int\limits_0^u \mathbf{E}\int\limits_0^u \mathbf{E}du\,du + \frac{e}{mc\omega\alpha}\int\limits_0^u \mathbf{p}_{\parallel 0}\mathbf{E}\,du;$$

$$z = \frac{e^2}{m^2c\omega^3\alpha^2}\int\limits_0^u\int\limits_0^u \mathbf{E}\int\limits_0^u \mathbf{E}\,du\,dudu + \frac{e}{m^2c\omega^2\alpha^2}\int\limits_0^u\int\limits_0^u \mathbf{p}_{\parallel 0}\mathbf{E}\,dudu;$$

$$\mathbf{p}_\perp = \frac{e}{c}\int\limits_0^u \mathbf{E}\,du + \mathbf{p}_{\perp 0};$$

$$\mathbf{r}_\perp = \frac{e}{mc\omega\alpha}\int\limits_0^u\int\limits_0^u \mathbf{E}\,dudu + \int\limits_0^u \frac{\mathbf{p}_{\perp 0}}{m\omega\alpha}\,du.$$

Были испытаны две модели волны.

$$\mathbf{E} = \{E_0\frac{d^2\mathrm{e}^{-u^2}}{du^2},\ 0,\ 0\}$$

и

$$\mathbf{E} = \{E_x,\ 0,\ 0\},$$

$$E_x = \begin{cases} E_0(1-u^2), & -1<u<1;\\ 0, & u\leqslant -1, u\geqslant 1.\end{cases}$$

Программа Maple, которая интегрировала уравнения движения для второго случая показана ниже.

```
> restart; E:=u->piecewise(u<1 and u>-1, E0*(1-u^2), 0);
```

$$E := u \to \mathrm{piecewise}(u<1\ \mathbf{and}\ -1<u),\ E0\,(1-u^2),\ 0)$$

```
> E(u);
```

$$\begin{cases} E0\,(1-u^2) & u-1<0\ \mathbf{and}\ -1-u<0\\ 0 & \mathbf{otherwise}\end{cases}$$

```
> px:=simplify(e*int(E(u), u)/omega);
```

$$px := \begin{cases} 0 & u\le -1\\ -\frac{1}{3}\frac{e\,E0\,(-3\,u-2+u^3)}{\omega} & u\le 1\\ \frac{4}{3}\frac{e\,E0}{\omega} & 1<u\end{cases}$$

```
> x:=simplify(int(px, u)/m/omega/alpha);
```

$$x := \begin{cases} 0 & u\le -1\\ -\frac{1}{12}\frac{e\,E0\,(-6\,u^2-3-8\,u+u^4)}{m\,\omega^2\,\alpha} & u\le 1\\ \frac{4}{3}\frac{e\,E0\,u}{m\,\omega^2\,\alpha} & 1<u\end{cases}$$


```
> pz:=simplify(e^2*int(E(u)*int(E(u), u), u)/m/c/omega^2/alpha);
```

$$pz := \begin{cases} 0 & u \le -1 \\ \dfrac{1}{18}\,\dfrac{e^2\,E0^2\,(9\,u^2+4+12\,u-6\,u^4-4\,u^3+u^6)}{m\,c\,\omega^2\,\alpha} & u \le 1 \\ \dfrac{8}{9}\,\dfrac{e^2\,E0^2}{m\,c\,\omega^2\,\alpha} & 1 < u \end{cases}$$

```
> z:=simplify(int(pz, u)/m/omega/alpha);
```

$$z := \begin{cases} 0 & u \le -1 \\ \dfrac{1}{630}\,\dfrac{e^2\,E0^2\,(105\,u^3+33+140\,u+210\,u^2-42\,u^5-35\,u^4+5\,u^7)}{m^2\,\omega^3\,\alpha^2\,c} & u \le 1 \\ \dfrac{8}{315}\,\dfrac{e^2\,E0^2\,(-9+35\,u)}{m^2\,\omega^3\,\alpha^2\,c} & 1 < u \end{cases}$$

Таким образом, Maple легко проводит вычисления с кусочно-гладкими функциями.

Решение показало, что для первой модели волны ускорения заряда не происходит, а после взаимодействия со второй волной заряд ускоряется. Была выдвинута гипотеза об ускорении заряда в ударных волнах, напряжённость которых удовлетворяет условию $\int\limits_{-\infty}^{\infty} \mathbf{E}\,du \ne 0$.

Дополнительные пакеты Maple

Maple содержит большой набор (37) весьма полезных библиотек для различных областей математики начиная от математической логики и статистики до тензорного исчисления и теории групп. Стандартные пакеты подключаются командой with и не могут быть изменены. К таким пакетам относятся plots, linalg (операции с матрицами), stats (статистика), tensor, PDEtools (работа с уравнениями в частных производных), student.

Существует так же изменяемый набор пакетов, которые подключаются командой readlib. Эти пакеты может создавать и сам пользователь. Ниже приведена программа, которая вычисляет несколько членов двумерного ряда Тейлора и затем генерирует текст программы для языка C.

```
> restart; readlib(mtaylor): readlib(C):
> t:=mtaylor(tan(x^2-y^2), [x,y], 7);
```

$$t := x^2 - y^2 + \frac{1}{3}\,x^6 - y^2\,x^4 + y^4\,x^2 - \frac{1}{3}\,y^6$$

```
> C(t, optimized);
```

```
t1 = x*x;

t2 = y*y;

t3 = t1*t1;

t7 = t2*t2;

t11 = t1-t2+t3*t1/3.0-t2*t3+t7*t1-t7*t2/3.0;
```

Сравнение Maple с аналогичными средами

Сравнивались версии программ

Maple V Release 5.1, MathCAD 8.1, Mathematica 4.0

Параметры	**Maple**	**MathCAD**	**Mathematica**
Импорт	-	Большие возможности настройки ввода	-
Экспорт	HTML, LaTeX	RTF, HTML, E-Mail	HTML, TeX, E-Mail
Помощь	англ.	англ./рус.	англ.
Визуализация	Хорошо	Отлично	Плохо
Настройка	Хорошо	Не требуется	Очень сложно
Особенности			Возможность работы со звуком

Программы сравнивались прежде всего по возможностям импорта-экспорта файлов. Сразу отметим, что ни одна из этих программ не может ни импортировать, ни экспортировать внутренний формат двух остальных программ. MathCAD позволяет сохранять файлы в формате RTF, что означает слияние с популярным текстовым процессором Microsoft Word.

Синтаксис языка системы Maple, для знакомых с языками высокого уровня (C, Pascal и, конечно, Fortran) более понятен по сравнению с Mathematica. MathCAD все операции производит, вставляя шаблоны математических объектов в любое место листа при помощи палитр или через меню. Работа из "командной строки" полностью исключена. Вместе с тем, в Maple и Mathematica существуют палитры шаблонов, которые вставляют команды с незаполненными аргументами в позицию курсора. Опыт показывает, что палитры удобны только при освоении синтаксиса команд.

При вводе сложных выражений Maple выгодно отличается от Mathematica тем, что может поле ввода представить в удобном для просмотра виде, благодаря чему легко найти ошибку.

Литература

1. Гердт В. П., Тарасов О. В., Широков Д. В. Аналитические вычисления на ЭВМ в приложении к физике и математике//УФН. 1980. Т. 130, вып. 1. С. 113.

2. Аналитические вычисления на ЭВМ и их применение в теоретической физике: Тр. совещ. Дубна: ОИЯИ, 1983. С. 260.

3. Аналитические вычисления на ЭВМ и их применение в теоретической физике: Тр. совещ. Дубна: ОИЯИ, 1985. С. 420.

4. Климов Д. М., Руденко В. М. Методы компьютерной алгебры в задачах механики. – М.: Наука, 1989.

5. Гурин Н. И., Скоморохов А. Г. Аналитические вычисления в системе REDUCE. – Мн.: Наука и техника, 1989.

6. Демидович Б. П. Марон И. А. Основы вычислительной математики. М., 1970.

7. Справочник по специальным функциям. Под ред. Абрамовица М. и Стиган И. – М.: Наука, 1979.

8. Коваль П. Г., Копытов Г. Ф. Ондуляторное излучение//Труды ФОРА, 1999, №4, С. 81 – С. 89.

9. Ландау Л. Д., Лифшиц Е. М. Теоретическая физика Т. 2. – М.: Наука, 1988.

Using a computer system Maple in the theoretical physicits

P. G. Koval

In the article is considered several examples of using a computer system Maple at calculations on the theoretical physicist. Brought fragments of concrete programs. Conduct comparison with other similar software.

